

# Excel2021 使用教程

## 第一讲 认识 Excel

### 1、公式必须以等号开头

如果大量输入公式，想省略等号直接得出结果：文件——选项——高级——

Lotus 兼容性公式——勾选转换 Lotus1-2-3 公式

### 2、EXCEL 能做些什么？

数据存储→数据处理→数据分析→数据呈现

### 3、常用文件类型

①xls/xlsx 工作簿文件

②xlw 工作区文件：视图——新建窗口（变成两个窗口）——全部重排（同一文件的两个表可以进行比较，修改其中的一个，另一个也会改变）——保存工作区（其实就是保存表格的布局）

★Excel 从 2013 版本开始已经没有“保存工作区”功能

★工作区就是一个超链接，快捷方式，不存储数据

### 4、工作簿、工作表、单元格

①工作簿中包含多个工作表，在表格下侧点击可插入新的工作表

②右击新的工作表，可更改工作表名称和颜色

③点击第一个工作表—按住 SHIFT 键—点击最后一个工作表—右击—插入/删除多个工作表

④在 3、4 行中间插入一行，选中第 4 行—右击—插入，若选择多行插入时，便可插入多行。列同理。

⑤整列交换，按住 SHIFT 键，鼠标放在边框线上，移动

⑥调整宽，可手动左右拉，或者双击，自动调整到合适宽度。选择多列一起手动调整，则每列宽度都一样。

⑦如果想从表的任意一个位置到达表格最后一行，选中任一表格，鼠标放在表格边框线上，双击。

## 5、使用小工具

### ①冻结窗格

视图—冻结窗格—冻结窗格（以当前单元格在哪为依据，冻结当前单元格以上与左侧的所有格）

### ②填充柄

选中一个单元格—点击—鼠标放在右下角，变成“+”—鼠标点击左键向下来，拖拽出的单元格都一样。写一个 1，在下方单元格写一个 2，以相同方法拖拽 2，就出现了 1、2、3、4、5…

输入“ctrl:”就会输入今天日期。

同时按住 ctrl、shift 和冒号键就会输入现在的电脑时间 6:49

★直接拖拽是序列，按住 ctrl 键拖拽就是复制；直接拖拽是复制，按住 ctrl 键拖拽就是序列。

★日期选中鼠标右键拖拽，然后松开，就可以选择以何种方式填充。

★自定义拖拽顺序：文件—选项—高级—在常规模块中点击编辑自定义列表—点击自定义序列中的新序列—输入序列—添加—确定

## 6、快速选择第二到第九百行，可以使用名称框，输入 2:900

# 第二讲 单元格格式设置

## 1、使用单元格格式工具美化表格

选择单元格——点击开始菜单——点击对齐方式右下角的小方格——弹出“设置单元格格式

“窗口


① “对齐”选项卡，设置文字对齐方式

多行单元格每行都合并，而列不合并：选中多行单元格——开始菜单——合并后居中——跨越合并


② “边框”选项卡，设置单元格边框

单元格内输入文字需要另起一行，需要按住 Alt 键+回车键再输入第二行的内容


★制作常见的两个项目的斜线表头：上面的内容先写——另起一行写下面的内容——在单元格内用边框画斜线——全部设置为左对齐——上面的内容敲打空格挪到右边



设置单元格格式



★制作三个项目的斜线表头：需要画斜线（缺点：斜线无法随着单元格进行移动或者大小变换，每次都需要人为拖动）



③ “字体”选项卡，设置字体颜色

④ “填充”选项卡，设置单元格背景颜色

★使用格式刷可完整复制表格样式应用至新的表格，双击格式刷可以在不同的地方循环使用格式刷（Esc 键取消）

## 2、单元格数字格式

类型	原格式	转变后的格式
数值	-25636	-25,636
货币	10000	¥10,000.00
会计专用	1555	\$ 1,555.00
日期	39814	2009年1月1日
时间	0.6980556	下午4时45分12秒
百分比	0.11	11.00%
分数	0.1	0
科学计数	1.2E+14	1.2E+14
文本	2422	2422
特殊	25368	二万五千三百六十八

★Excel 采用 1900 年作为元年，1 代表 1900 年 1 月 1 日，输入日期 2022/6/19 如果改为数字格式则为 44731，表示该天距 1900 年 1 月 1 日有 44731 天

★Excel 中数字永远都是数字，使用不同单元格数字格式可以转换为不同的显示样式

### ★特殊格式中可以将数字转换为中文大写数字

★批量设置自定义单元格格式：选中要修改的单元格——“设置单元格格式”——“自定义”——选择类似样式，直接在后面添加需要内容，例如“元”

**优势：**更改了格式之后，数值依然能够进行公式运算，且运算的结果和自定义格式相同



★使用自定义格式隐藏文本：选中需要隐藏的文本——“设置单元格格式”——自定义输入“; ; ; ”



★使用自定义修改日期的显示格式:

yyy/mm/dd: 2022/06/19

www-mm-dd: 2022-06-19

dd-mmm-yyyy: 19-Jun-2022

dd-mm-mm-vvv: 19-June-2022

yyyy-mm-dd: 2022-06-Sun

yyyymmdd: 2022-06-Sunday

aaa: 日

aaaa: 星期日

周 aaa: 周日

★身份证填写方式：单元格格式设置为文本

★文本改为数值：文本不能通过设置单元格格式转换为其他类型的格式，选中格式为文本的单元格，点击前面的黄色感叹号图标，选择转换为数字

如果大量文本与数字格式的数据混在一起想全部转换为数字，则全部乘 1


单元格填入数字 1。复制该单元格——选中所有需要转换的单元格——右键选择性粘贴——

—运算点选乘——确定

### 3、使用“分列”工具

数据菜单——数据工具——分列——选择“分隔符号”或“固定列宽”——“下一步”——分隔符号选择——完成

★日期格式更改：2022/6/19 格式是无法直接通过设置单元格格式转换为“2022 年 6 月 19 日”的，必须要使用分列工具，在列数据格式中选择常规或者日期格式，然后再到设置单元格格式中进行日期格式的更改



## 第三讲 查找、替换与定位

### 1、查找与替换

①按值查找：精确匹配

开始菜单——查找和选择——替换——输入查找与替换内容——勾选单元格匹配

②按格式查找

开始菜单——查找和选择——替换——格式——从单元格选择格式

★只要有一个格式被识别，就能替换，并不是所有格式都必须选择设置上

③模糊查找（认识通配符\*?~）

★\*：星号表示通配符，为任意值，无论几个值都可以用星号表示

★?：英文状态下的问号，为任意值，但是只表示一个值，大多数时需要勾选单元格匹配

★~：小波浪线~放在\*或?的前面，意思是让后面的字符不生效，用在替换通配符的本身

★ctrl+F：键盘上查找的快捷键

★ctrl+H：键盘上替换的快捷键

### 2、定位

①通过名称框定位单元格及区域位置

定位 A 列 10000 行，名称框：A10000

全选 A、B 列 9000-10000 行，名称框：A9000:B10000

全选 9000-10000 行，名称框：9000:10000

②定义名称

选择所需的单元格区域——名称框内命名，再按回车键——以后查找该区域直接搜索名字

③使用定位条件选择注释、批注、空值、图片

开始菜单——查找和选择——转到——填写引用位置——回车

——定位条件——注释

### ★注释：形式类似于 2010 版的批注

审阅菜单——注释——新建注释/编辑注释/显示所有注释

点中注释框——在快速访问工具栏点击编辑形状——更改形状

点中注释框——鼠标右键——设置批注格式——颜色与线条——在颜色下拉框中选择填充效果——图片——选择图片——确定

★批注：给单元格做小备注页，只有鼠标悬停的时候才能看到备注内容

选中单元格——鼠标右键——新建批注

鼠标悬停出现批注，点击右下角编辑，可以直接修改批注


选中单元格——鼠标右键——删除批注（选中全部单元格可以批量删除批注）

审阅菜单——批注栏中显示批注——所有批注均显示出来

★插入菜单——形状——随机画个矩形——选中该矩形——形状格式菜单——在编辑形状的位置鼠标点击右键——添加到快速访问工具栏

★Excel2021 版本暂时未找到更改批注形状与批注添加图片的方法

取消合并单元格，空值批量填充最上方单元格的内容：



第一个空值里面填入“=J7”（第一行填有北京单元格的位置）或者输入=后按住键盘上的↑或者=后鼠标点上格，再按住 ctrl 键加回车键。

★在 Excel 中选择图片删除

所有图片删除：定位条件——对象——选中所有图片——delete 键

部分图片删除：查找和选择——选择对象——自由选择图片

## 第四讲 排序与筛选

### 1、排序

#### ①简单排序

选中一个单元格/选中整张表格——排序与筛选——升序/降序

#### ②多条件排序

选中一个单元格——排序和筛选——自定义排序——主要关键字——排序依据——次序——+添加条件——确定

#### ③按颜色排序

选中一个单元格——排序和筛选——自定义排序——主要关键字——排序依据：单元格颜色——次序：选择单元格颜色——在顶端/在底端

#### ④自定义排序次序

选中一个单元格——排序和筛选——自定义排序——主要关键字——排序依据——次序：

自定义序列——新序列——输入序列——确定

#### ⑤利用排序插入行

先批量制作出要插入的行——在表格旁边单独起一行——上方数据从 0、1、2……——下方要插入的数据从 1.5、2.5……——最后按照数字进行全部排序

★打印表格，每一页第一行都是标题行：

页面布局菜单——页面设置右下角箭头——工作表——顶端标题行选中标题行所在的位置——确定——打印：每一页的标题都是同一个

## 2、筛选

### ①使用筛选

开始菜单——排序与筛选——筛选——选择所需数据/数字筛选/文本筛选（可使用\*和?两个通配符）

★若复制粘贴筛选之后的数据出现全部未筛选的数据，则需要选择查找和选择——定位条件——点选可见单元格——然后再去复制和粘贴

### ②在筛选中使用多个条件

### ③高级筛选

A、筛选不重复值

数据菜单——排序与筛选模块——高级——点选将筛选结果复制到其他位置——列表区域：填写筛选的区域——复制到：选择一个单元格——勾选“选择不重复的记录”——确定

B、在高级筛选中使用常量条件区域

★条件区域：必须带表头，条件写在同行表示“且”，条件写在不同行表示“或”，有几行表示有多少个或。

部门	科目划分	部门	科目划分	部门	发生额
一部门	邮寄费	一部门	邮寄费	一车间	>3000
				二车间	>10000

C、在高级筛选中使用函数公式条件区域

★条件区域一旦使用公式，则不需要写表头，作为空值放置

★迅速全选表格的方法：点击表格最左上角的单元格——按住 ctrl+shift 两个键——敲下键盘上的“→”键，选择了第一行全部的单元格——再敲下键盘上的“↓”键，全部表格都已经选中。

# 第五讲 分类汇总、数据有效性

## 1、分类汇总工具

### ①认识分类汇总

数据菜单——排序（升序/降序）——分类汇总——分类字段——汇总方式——选定汇总项——确定

★左边出现的 1，2，3 分别表示总计、汇总统计、详细的全部数据

★删除分类汇总：分类汇总——全部删除

### ②使用分类汇总前先排序

### ③分类汇总的嵌套

多个分类依据则数据分主次排序后使用多次的分类汇总（分类汇总要勾选掉替换当前的分类汇总）



#### ④复制分类汇总的结果区域

选择分类汇总的结果区域——开始菜单——查找与选择——定位条件——可见单元格——  
ctrl+c——ctrl+v

可见单元格快捷键：Alt+;

#### ⑤使用分类汇总批量合并内容相同的单元格（不一样的内容）

排序——分类汇总——选择你需要合并相同内容的项目作为分类字段——汇总方式：计数——确定——选择出现分类汇总的那一列——查找与选择——定位条件——勾选空值——合并单元格——分类汇总——全部删除——选中合并单元格的那一列——使用格式刷在需要合并内容相同的单元格

### 2、设置数据有效性（给数据设定规则）

Excel2021 版的数据有效性改为数据验证

#### ①设置整数数据有效性

数据菜单——数据验证——数据验证——验证条件允许选择整数——数据：介于——最小值/最大值

#### ②设置文本长度数据有效性

数据菜单——数据验证——数据验证——允许：文本长度——数据：等于——输入规定文本长度

#### ③设置序列数据有效性

数据菜单——数据验证——数据验证——允许：序列（固定输入某些值，只能选序列）——来源（手写输入，中间用英文逗号连接不同内容）

#### ④数据有效性的其他设置

##### A、输入法切换

数据菜单——数据验证——数据验证——输入法模式——关闭：英文模式；打开：中文模式

★需要更改电脑设置，让电脑语言栏既存在中文又存在英文，但是建议不要改

##### B、单元格信息

##### C、单元格信息保护

★数据验证可以输入公式运算：数据菜单——数据验证——数据验证——允许：自定义——公式

★数据验证通过自定义公式可以禁止数据更改并发出警告：数据菜单——数据验证——数据验证——允许：自定义——公式：输入不可能达成运算的公式例如=1>2——出错警告——停止/警告/信息：可输入标题和错误信息对更改人给予提示

## 第六讲 数据透视表应用

### 1、数据透视表

#### ①创建数据透视表

一列为一个字段；每行为一条完整的记录

选中任意一个表中单元格——插入菜单——数据透视表——确定——数据透视表选项——显示——勾选经典数据透视表布局（启用网络中的字段拖放）——确定——选择要添加到报表的字段：拖动到行/列/值字段中去

#### ②更改数据透视表汇总方式

在左上角第一个单元格上双击（求和项……）——弹出“值字段设置”命令框——更改计算类



型——确定

★在透视表的值字段中，双击某个单元格可以返回与之相关的数据表格

③数据透视表中的组合

点击任意一个单元格——右键——组合——分组

★行字段可以有多个组合，只需要拖动到合适位置就行，若出现分类汇总项，则双击标题，在分类汇总和筛选上勾选“无”

★透视表中的字段可以重复使用，例如既可以作为行字段也可以作为值字段

④汇总多列数据

值字段要进行多种不同的计算方法，可以拖拽多次

多个字段拖拽到相同区域时，数据必定会上下排列，把数据标题拉到汇总行（列字段），数据会变成平行排列，然后修改值字段的计算方式（点击标题单元格，选择计算类型）

★可直接在标题栏里更改标题，但不能与之前的存在的字段标题重复

⑤在透视表中使用计算

点中数据透视表任意单元格——数据透视表分析——字段、项目和集——计算字段——修改名称——公式：双击字段中所需公式的参数——确定

★若使用计算后，表格中出现错误，右键数据透视表选项——布局和格式——勾选并填写：对于错误值，显示——确定

⑥利用筛选字段自动创建工作表

报表筛选字段就是一个筛选字段

如果想批量制作工作表，可以选择项目中任意一个单元格——插入——数据透视表——将需要批量制作的项目拖拽到报表筛选字段和值字段上——点中透视表任意一个单元格——数据透视表分析——选项——显示报表筛选页——确定——选中所有表格——使用复制粘贴空白单元格，删除数据透视表

# 第七讲 认识公式与函数

## 1、认识 EXCEL 公式

### ①运算符

算术运算符+ - \* / % & ^

&：连字符，将单元格中的文字/数字连接起来

+ - \* /：可以处理文本或数字，但文本经过处理会转换为数值

比较运算符= > < >= <= <>

### ②公式中的比较判断

比较运算符的结果，TRUE，FALSE

TRUE：1

FALSE：0

### ③运算符优先级

-	负号
%	百分比
^	求幂
*/	乘除
+ -	加减
&	文本连接

=,<,>,>=,<=,<>	比较
----------------	----

- ④单元格引用
  - 相对引用, A1
  - 绝对引用, \$A\$1
  - 混合引用, \$A1, A\$1
  - ★绝对引用的快捷键: 选中需要绝对引用的单元格的名称, 按下 Fn+F4 键, 自动转换为绝对引用, 再按一次转换为混合引用, 再按一次转换为相对引用, 循环切换
  - ★混合引用:
    - A\$1 向下拖拽, 单元格不会变化, 向两边拖拽, 只改变为 B\$1, C\$1…… (锁定 1)
    - \$A1 向下拖拽, 单元格变化为\$A2, \$A3……, 向两边拖拽, 单元格不会变化 (锁定 A)
    - \$这个符号在前意味着锁定

## 2、认识函数

- ①如何使用函数
  - 等号开头
  - 函数名在中间
  - 括号结尾
  - 括号中间写参数
- ②学习以下函数
  - SUM 求和
  - AVERAGE 求平均
  - MIN 求最小
  - MAX 求最大
  - COUNT 与 COUNTA 计数
  - RANK 排名
  - ★同样可以使用开始菜单中的求和符号下拉列表中的几种常见运算函数, 可以帮助自动选区, 在手动选区很麻烦的时候使用
- ③利用定位工具选择输入公式的位置
  - 选中单元格区域——查找和选择——定位条件——空值——自动求和工具符号

# 第八讲 IF 函数逻辑判断

## 1、使用 IF 函数

- ①IF 函数的基本用法
  - 函数语法: IF(logical\_test,[value\_if\_true],[value\_if\_false])
- ②IF 函数的嵌套
  - 例如: =IF(A1="理工","LG",IF(A1="文科","WK","CJ"))
  - =IF(D1="本地",E1+30,IF(D1="本省",E1+20,E1+10))
  - =IF(F1>=600,"第一批",IF(F1>=400,"第二批","落榜"))

专业类后加专业代号	本地学生 总分为原始分加 30	录取情况	
理工 对号 LG	本省学生 总分为原始分加 20	600分含600	显示 第一批
文科 对号 WK	外省学生 总分为原始分加 10	400-600 含400分	显示 第二批
财经 对号 CJ		400分以下	落榜

- ③如何尽量回避 IF 函数的嵌套

可以使用多个 IF 函数相加+或连接&, 减少 IF 函数的嵌套

数值: 若不符合返回 0

文字: 若不符合返回空值 ("")

★多个字段逻辑判断返回值可以使用 vlookup 函数

④用 IF 函数处理运算错误 Iserror 函数

在工作表中需要回避错误 (将错误值返回为 0 或空值): =IF(ISERROR(C6/D6),0,C6/D6)

★ISERROR 函数: 最全面的错误函数, 若错误返回为 TRUE, TRUE 可以当作 1

## 2、AND 函数与 OR 函数

①AND 函数: 表示“且”的关系

例如: =IF(AND(A11="男",B11>=60),1000,0)

②OR 函数: 表示“或”的关系

例如: =IF(OR(B11<=40,B11>=60),1000,0)

★写 IF 函数前先理清思路很重要!

例如: =IF(OR(AND(A11="男",B11>=60),AND(A11="女",B11<=40)),1000,0)

# 第九讲 COUNTIF 函数

## 1、使用 COUNTIF 函数

①Count 函数

Count 专门用来点数 (只能数数字)

②Countif 函数语法

分类计数: 涵义是数数, 带一个条件逻辑

Countif(range,criteria)

Range: 计数范围

Criteria: 计数的类型, 数什么值

③Countif 函数计算数值区间

例如: =COUNTIF(B11:B15,">=60")

在进行数字比较 (数值区间) 的时候, 需要使用""将判断要求包起来, 因为数值区间既不属于数字也不属于公式, 只能将它看作一个字符串 (文本)

④Countif 函数超过 15 位字符时的错误

=COUNTIF(A18:A24,A18&"\*")

解决方法: 在后面加上"&"\*", 人为增加 Excel 中默认的 15 位字符的长度

## 2、常见应用示例

①在数据区域中寻找重复数据

例如: =IF(COUNTIF(I:I,F10)=0,"未体检","已体检")

②在数据有效性 (数据验证) 中使用 Countif 函数

将 H 列设置为禁止输入重复数据: 数据菜单——数据验证——数据验证——设置——允许: 自定义——公式: =countif(H:H,H1)<2——确定

③在条件格式中使用 Countif 函数

选中表格中的区域——开始菜单——条件格式——新建规则——使用公式确定要设置格式的单元格——为符合此公式的值设置格式: =COUNTIF(I:I,F10)=0——格式——确定

## 3、COUNTIFS 函数

多条件筛选计数

例如：=COUNTIFS(D:D,A1,E:E,B1)

## 第十讲 SUMIF 函数

### 1、使用 SUMIF 函数

#### ①SUMIF 函数语法

SUMIF 函数帮助找到某个区域当中符合条件的值并且把它们后面的某一列相加

SUMIF(range, criteria, [sum\_range])

Range: 筛选条件所在的区域

Criteria: 筛选条件

[sum\_range]: 相加数值的区域

例如：=SUMIF(D:D,C6,E:E)

#### ②Sumif 函数计算数值区间

例如：=SUMIF(D:D,">=500")

★第三参数省略：若筛选条件区域和相加数值区域是一样的，可省略不写相加数值区域

#### ③SUMIF 函数超过 15 位字符时的错误

例如：=SUMIF(A:A,D6&"\*",B:B)

#### ④关于第三参数简写时的注意事项

SUMIF 是容错率极高的函数，只要数值区域是规律性的，都可以简写第三参数，即选中该列的单元格，函数默认计算整列

A	B	C	D	E	F	G	H	I	J	K	L	M
科目划分	发生额	科目划分	发生额	科目划分	发生额	科目划分	发生额	科目划分	发生额		科目划分	发生额
邮寄费	5.00	邮寄费	150.00	交通工具消耗	600.00	手机电话费	1,300.00	公积金	15,783.00		邮寄费	1371
出租车费	14.80	话费补	180.00	采暖费补助	925.00	出差费	1,328.90	抵税运费	31,330.77		独子费	130
邮寄费	20.00	资料费	258.00	招待费	953.00	工会经费	1,421.66	办公用品	18.00		过桥过路费	1130
过桥过路费	50.00	办公用品	258.50	过桥过路费	1,010.00	出差费	1,755.00	出差费	36.00		手机电话费	1800
运费附加	56.00	养老保险	267.08	交通工具消耗	1,016.78	出差费	2,220.00	招待费	52.00			
独子费	65.00	出租车费	277.70	邮寄费	1,046.00	招待费	2,561.00	招待费	60.00			
过桥过路费	70.00	招待费	278.00	教育经费	1,066.25	出差费	2,977.90	独子费	65.00			
出差费	78.00	手机电话费	350.00	失业保险	1,068.00	出差费	3,048.40	出差费	78.00			
手机电话费	150.00	出差费	408.00	出差费	1,256.30	误餐费	3,600.00	招待费	80.00			
邮寄费	150.00	出差费	560.00	修理费	1,260.00	出差费	6,058.90	其他	95.00			

例如：计算整个工作表中的邮寄费，=SUMIF(A:I,L3,\$B\$1)

★一定要保证两列数据开始的位置相同，即 A1 对应 B1，不然计算会错位

#### ⑤在多列中使用 Sumif 函数

#### ⑥使用辅助列处理多条件的 sumif

辅助列	A	B	C	D	E ↓	F	G	H	I	J	K	L
一车间邮寄费		01	29	凭证号数	部门	科目划分	发生额					
一车间出租车费		01	29	记-0021	一车间	出租车费	14.80					
二车间邮寄费		01	31	记-0031	二车间	邮寄费	20.00				部门	科目划分
二车间过桥过路费		01	29	记-0022	二车间	过桥过路费	50.00				一车间	邮寄费
二车间运费附加		01	29	记-0023	二车间	运费附加	56.00				二车间	独子费
财务部独子费		01	24	记-0008	财务部	独子费	65.00				二车间	过桥过路费
二车间过桥过路费		01	29	记-0021	二车间	过桥过路费	70.00				二车间	手机电话费

辅助列：=E2&F2——发生额：=SUMIF(A:A,J5&K5,G:G)

#### ⑦Sumifs 函数

=SUMIFS(sum\_range, criteria\_range1, criteria1, [criteria\_range2, criteria2], ...)

SUMIFS 与 SUMIF 参数的顺序不一样，SUMIFS 先规定求和范围，后面再写筛选条件

例如：=SUMIFS(G:G,E:E,J5,F:F,K5)

#### ⑧复习数据有效性（数据验证）

	A	B	C	D	E	F	G
1	库存表					出库单	
2	彩盒	44855			日期	产品	数量
3	宠物用品	48595				彩盒	
4	服装	6902				彩盒	
5	警告标	48619					
6	睡袋	68462					
7	睡袋	147161					

给数量设定数据有效性（出库数量总计必须小于或等于库存数量）：

数据菜单——数据验证——数据验证——自定义——公式：  
=SUMIF(F:F,F3,G:G)<=SUMIF(A:A,F3,B:B)——确定

★SUMIF(F:F,F3,G:G)：对出库数量进行求和

SUMIF(A:A,F3,B:B)：对库存数量进行求和（每个项目不重复，仅有一个对应的值）

## 第十一讲 vlookup 函数

### 1、使用 VLOOKUP 函数

#### ①vlookup 函数语法

=VLOOKUP（查找值，包含查找值的范围，包含返回值的范围中的列号，近似匹配（TRUE）或精确匹配（FALSE））

★查找值的范围：必须保证查找值所在的一列在最左侧，范围是从查找值到返回值两列之间的表格，区域要使用绝对引用

★模糊匹配 TRUE：1；精确匹配 FALSE：0

#### ②vlookup 中使用通配符

查找值没有完全精确的情况下需要使用通配符

例如：=VLOOKUP(A2&"\*",数据源!B:E,4,0)

#### ③vlookup 模糊查找

模糊查找仅仅只用于数值区间查找，且只会查找小于或等于查找值的最大值

等级划分	提成比例				
0	5.00%	销售人员	汇总	提成比例	提成金额
2000000	10.00%	白桦	3683855	0.1	368385.5
4000000	15.00%	冯文	13265432	0.3	3979629
6000000	20.00%	蒋波	16409190	0.3	4922757
8000000	25.00%	刘辉	9677803	0.25	2419451
10000000	30.00%	熊牧	8330938	0.25	2082734
		张明	9101320	0.25	2275330
		赵温江	21848837	0.3	6554651
		郑浪	2270090	0.1	227009

★数值区间需要从小到大排序

#### ④使用 isna 函数处理数字格式引起的错误

查找项为数值，而查找值范围为文本，则需要在查找项后面连接"&"", 强行将数值转换为文本

查找项为文本，而查找值范围为数值，则需要在查找项后面\*1/+0（也可以在查找项的前面写--），强行将文本转换为数值

如果出现文本与数值非常混乱的情况，需要使用 IF 函数和 isna 函数进行判断



17	编号	发货地点	出货数量			根据编号查找出货数量
18	1001	上海	95			
19	1002	上海	64			
20	1003	上海	103			
21	1004	上海	11			
22	1005	上海	9			

=IF(ISNA(vlookup(F20+1,\$A\$17:\$C\$22,3,0)),vlookup(F20&"", \$A\$17:\$C\$22,3,0),  
vlookup(F20+1,\$A\$17:\$C\$22,3,0))

★推荐直接统一全部的查找项，统一为数值或文本

#### ⑤ Hlookup 函数

Hlookup 在行里查找值，一列为一个数据，与 vlookup 的方向不同

=HLOOKUP (查找值, 包含查找值的范围, 包含返回值的范围中的行号, 近似匹配 (TRUE) 或精确匹配 (FALSE))

#### 2、使用 XLOOKUP 函数 (EXCEL2021)

=XLOOKUP(lookup\_value, lookup\_array, return\_array, [if\_not\_found], [match\_mode], [search\_mode])

return\_array: 要返回的数组或区域

[if\_not\_found]: 如果找不到有效的匹配项，请返回你提供的 [if\_not\_found] 文本

[match\_mode]: 指定匹配类型:

0: 完全匹配。如果未找到，则返回 #N/A。这是默认选项。

-1: 完全匹配。如果没有找到，则返回下一个较小的项。

1: 完全匹配。如果没有找到，则返回下一个较大的项。

2: 通配符匹配，其中 \*, ? 和 ~ 有特殊含义。

[search\_mode]: 指定要使用的搜索模式:

1: 从第一项开始执行搜索。这是默认选项。

-1: 从最后一项开始执行反向搜索。

2: 执行依赖于 lookup\_array 按升序排序的二进制搜索。如果未排序，将返回无效结果。

-2: 执行依赖于 lookup\_array 按降序排序的二进制搜索。如果未排序，将返回无效结果。

	A	B	C	D
1		Emp ID	雇员姓名	部门
2		8389	柏隼	财务
3				
4		Emp ID	雇员姓名	部门
5		4390	赵强	市场营销
6		8604	宋臻	销售
7		8389	柏隼	财务
8		4937	康露	会计
9		8299	茅彩	运营
10		2643	王鐸	执政
11		5243	游皓	销售员
12		9693	曹卿	财务
13		1636	林耀卉	会计
14		6703	任月英	市场营销

## 第十二讲 Match+Index 函数

### 1、MATCH 与 INDEX 函数

Vlookup 仅能从左侧数据查询右侧数据，无法反向查找

#### ① 函数语法

Match 与 index 两个函数连用，实现类似于 vlookup 的反向查找功能，match 负责查找，index 负责引用

MATCH(lookup\_value, lookup\_array, [match\_type])

INDEX(array, row\_num, [column\_num])

例如：=index(数据源!A:A,match(查询!A2,数据源!B:B,0))

②match+index 与 vlookup 函数比较

③使用 match 与 vlookup 函数嵌套返回多列结果

A	B	C	D	E	F	G	H	I	J	K
客户ID	公司名称	联系人姓名	联系人头衔	地址	城市	地区	邮政编码	国家	电话	传真
AABBCCD	三川实业有限公司	刘小姐	销售代表	大寨路 50	天津	华北	343567	中国	(030) 300743	(030) 30765452
ANATR	东南实业	王先生	物主	承德西路 80	天津	华北	234575	中国	(030) 355541	(030) 35553744
ANTON	坦森行贸易	王炫皓	物主	黄台北路 78	石家庄	华北	985060	中国	(0321) 55536	

B3		=VLOOKUP(\$A3,数据源!\$A:\$K,MATCH(返回多列结果!B\$2,数据源!\$A\$1:\$K\$1,0),0)									
A	B	C	D	E	F	G	H	I	J	K	
1		2	5	3	10	6					
2	客户ID	公司名称	地址	联系人姓名	电话	城市					
3	CHOPS	浩天旅行	白广路 31	方先生	(030) 300	天津					
4	ERNSH	正人资源	临江东街	谢小姐	(0571) 76	深圳					
5	HANAR	实翼	永惠西街	谢小姐	(0211) 55	南昌					
6	ANTON	坦森行贸易	黄台北路	王炫皓	(0321) 55	石家庄					
7	LAUGB	和福建设	创业西路	刘先生	(030) 155	天津					
8	LAZYK	春永建设	劳动辅路	王先生	(0571) 35	深圳					
9	LEHMS	幸义房屋	七一路 85	刘先生	(069) 202	南京					

Match 函数用来定位查找值所在的列，返回该值的列号，相当于查找“公司名称”在该列当中的第几个

④认识 column 函数

=VLOOKUP(\$D4,数据源!\$A:\$K,COLUMN()-3,0)						
D	E	F	G	H	I	J
	2	3	4	5	6	7
客户ID	公司名称	联系人姓名	联系人头衔	地址	城市	地区
CHOPS	=VLOOKUP(\$D4,数据源!\$A:\$K,COLUMN()-3,0)					
ERNSH	VLOOKUP(lookup_value,table_array,col_index_num,[range_lookup])					
HANAR						
ANTON						
LAUGB						
LAZYK						
LEHMS						

column 函数里面什么都不写默认返回所在单元格的列数

\$D4 表示向下拖拽，数字会发生变化但是 D 列永远不会变

⑤使用函数引用照片

公式菜单——定义名称——输入名称——引用位置：=INDEX(引用图片位置（绝对引用），MATCH(查找项（绝对引用），查找项所在的范围（绝对引用），0))——确定

文件菜单——选项——自定义功能区——从下列位置选择命令：所有命令——照相机——添加>>（任意位置都行，只要你找得到）——确定——点中照相机，选择任一单元格，出现图片——点中图片，在编辑栏中写入开始在公式中定义得名称——回车

★按住 alt 键拉动图片，图片会自动贴住单元格边缘

## 第十三讲 Word 邮件合并

### 1、简单的邮件合并



### ①批量生成多个文档

Word 中的邮件菜单——开始邮件合并——邮件合并分布向导——选择文档类型——开始文档——选择收件人：使用现有列表——通过浏览关联表格——撰写信函——其他项目——把表格中的部分信息关联信函内容——预览信函——合并到电子邮件/打印机/新文档

### ②利用 word 发送邮件

前面和批量生成多个文档的操作步骤相同，在合并到电子邮件时可以直接发送

### ③每页显示多条记录

类似试卷的格式

邮件菜单——开始邮件合并——邮件合并分布向导——文档类型：目录——选择开始文档——选择收件人：选择现有列表——浏览关联表格——选取目录——其他项目

### ④邮件合并后得数字格式处理

数字格式\#"#,###0"：表示带千分位分割符的数字，在域代码中写这些内容，更改数字格式用\#"#"

日期格式\@"M/d/yyyy"（M 需要大写）：更改日期格式用\@"#"

★域代码：快捷键 ALT+F9，格式更改后需要快捷键更新

## 第十四讲 日期函数

### 1、认识时间和日期

#### ①回顾日期格式

#### ②时间格式

对于 excel 中的数字而言，整数代表天，小数点后面的值为时间（即一天的 24 小时中这个时间点所代表的值）

#### ③基本的时间与日期运算

时间相减会得到<1 的小数，换算为分钟需要乘以 24 和 60

时间减去分钟需要将分钟换算为时间，分钟为>1 的整数，需要分钟/24/60

★单位必须要统一才能加减乘除，时间（小数）与日期（整数）必须要相互换算才能进行运算

### 2、日期函数

#### ①year、month、day 函数

这三个函数可以分开一个日期的年、月、日

#### ②date 函数

Date 函数合并年、月、日三个值

=date(year,month,day)

例如：

日期中月份的相加：=date(year(B5),month(B5)+C5,day(B5))

任意给一个日期，求这个日期所在月份的最后一天（可以把这个月份的下一月份的第一天减去一天就是这个月份的最后一天）：=date(year(B5),month(B5)+1,1)-1

或者 0 号也是表示上个月的第一天：=date(year(B5),month(B5)+1,0)

求本月的天数（先求本月的最后一天是哪一天，再看那一天是几号，有几号表明该月的天数）：=day(date(year(B5),month(B5)+1,0))

★date 函数一定会给予正确的日期值，会自动完成进位

#### ③datedif 函数

计算日期间隔 (excel 隐藏函数)

Datedif (开始日期, 结束日期, 类型)

“Y”时间段中的整年数

“M”时间段中的整月数

“D”时间段中的天数

“MD”日期中天数的差。忽略日期中的月和年

“YM”日期中月数的差。忽略日期中的日和年

“YD”日期中天数的差。忽略日期中的年。

例如：求间隔年份：=datedif(B5,C5,"y")

④weeknum

计算某个日期是一年的第几周

Return\_type: 1 星期从星期日开始。星期内的天数从 1 到 7 记数

2 星期从星期一开始。星期内的天数从 1 到 7 记数

WEEKNUM(serial\_number,[return\_type])

serial\_number: 该日期不能是文本格式

⑤weekday

计算某个日期是周几

WEEKDAY(serial\_number,[return\_type])

例如：="第"& WEEKNUM(B17,2)&"周第"&WEEKDAY(B17,2)&"天"

⑥text 函数

直接将日期转换为固定格式

TEXT(Value you want to format, "Format code you want to apply")

例如：将日期转换为星期几：=text(B3,"aaaa")

把假日期（如 20220629）转换为真正的日期格式：=text(B4,"0000-00-00")\*1

## 第十五讲 条件格式与公式

### 1、使用简单的条件格式

①为特定范围的数值标记特殊颜色

开始菜单——条件格式——突出显示单元格规则——设置为：自定义格式——在设置单元格格式中修改格式——确定

开始菜单——条件格式——突出显示单元格规则——文本包含……

②查找重复值

开始菜单——条件格式——突出显示单元格规则——重复值……

★清除条件格式：开始菜单——条件格式——清除规则

★删除重复值：数据菜单——高级筛选——勾选选择不重复的记录——确定

③为数据透视表中的数据制作数据条

在数据透视表中使用条件格式：选中数据透视表的数据区域——条件格式——数据条/色阶

★切片器（筛选数据的工具）：插入菜单——切片器——勾选分别阅读的内容——确定——可以分别出现不同的筛选项目——回到总表点击清除筛选器

### 2、定义多重条件的条件格式

①如何管理条件

隐藏错误单元格：开始菜单——条件格式——新建规则——只为包含以下内容的单元格设

置格式——单元格值改为错误——更改格式为白色的字体——确定

②创建多重条件

③多重条件的优先级问题

后做的会覆盖前面的条件格式，多重条件设定时，优先从最大值开始标记

### 3、使用公式定义条件格式

①条件格式中公式的书写规则

★条件格式只能对数据本身的值进行判断，无法标记与之相关联的单元格

开始菜单——条件格式——新建规则——使用公式确定要设置格式的单元格——填写判断命令，如：=D2>100

★条件格式选中区域后使用公式就类似于写完公式拖拽，要注意区分相对引用/绝对引用/混合引用，注意与第一个单元格的匹配

②使用 weekday 函数标记周末

③标记未来十五天的日期

★修改规则：开始菜单——条件格式——管理规则——编辑规则/删除规则

## 第十六讲 简单文本函数

### 1、使用文本截取字符串

①left 函数

从左向右取字符串：=LEFT(text,[num\_chars])

②Right 函数

从右往左取字符串：=RIGHT(text,[num\_chars])

③Mid 函数

从中间某一个位置取规定长度的字符串：=MID(text, start\_num, num\_chars)

★截取字符串 mid 函数：如果截取的字符串长度过大，依然只取原字符串原有的字符长度

④示例 使用 MOD 函数与文本函数，提取身份证号性别信息

十五位身份证最后一位是性别位，十八位身份证倒数第二位是性别位，在混合表中提取性别位：=RIGHT(LEFT(B13,17),1)

思路：将十八位身份证的性别位转换到最后一位，然后两种身份证都取最后一位

### 2、获取文本中的信息

文本函数：关心一个单元格里字符串的情况

①Len 与 Lenb 函数

文本长度函数，Len 函数：有几个字符显示个数；Lenb 函数：有几个字节显示个数（中文是双字节）

=LEN(text)

=LENB(text)

例如：求每个数据后面的中文单位：=RIGHT(A2,LENB(A2)-LEN(A2))

LENB-LEN 会得出中文单位的数量，283 元中字节数为 5，字符数为 4，相减等于 1；54 公斤中字节数为 6，字符数为 4，相减等于 2。从右往左取相应的字符数量即可得单位

②Find 函数

找一个字符在一个字符串里是第几位的函数

=FIND(find\_text, within\_text, [start\_num])

find\_text：找哪个字符串

within\_text: 在哪个区域里面找

[start\_num]: 从哪个字符位置开始找

例如: =left(F2,find("@",F2)-1)

=mid(F2,find("@",F2)+1,100)

★如果字符串中有多个@, find 只会找第一个@, 解决方法 (嵌套函数, 从第一个@之后开始找): =FIND("@",H11,FIND("@",H11)+1)

### 3、关于身份证

#### ①通过身份证前六位判断地区

通过文本处理函数得到的数字一定也是文本

=VLOOKUP(LEFT(B2,6)\*1,地区!A:B,2,0)

或=VLOOKUP(--LEFT(B2,6),地区!A:B,2,0)

#### ②通过身份证计算出年月日

=DATE(MID(B2,7,4),MID(B2,11,2),MID(B2,13,2))

或=TEXT(MID(B2,7,8),"0000-00-00")

#### ③通过身份证判断性别

#### ④身份证验证

身份证每一位乘以固定值, 然后将得出的数据相加除以 11, 求得的余数与另外一组值相互对应, 该对应值为该组身份证的最后一位, 若不匹配则为假

## 第十七讲 数学函数

### 1、认识函数

#### ①Round 函数、Roundup 函数、Rounddown 函数、Int 函数

Round 函数: 四舍五入函数

Roundup 函数: 无条件向上进位

Rounddown 函数: 无条件向下舍位

Int 函数: 直接取整数, 该函数所得数值永远小于原数值, 例如: int(-1.4)=-2

#### ②Mod 函数

Mod 函数: 求余数

=MOD(number, divisor)

Number: 被除数

Divisor: 除数

#### ③Row 函数与 Column 函数

Row 函数: 行, 求单元格的行数, 若 Row()则公式在第几行返回第几行的行号

Column 函数: 列, 求单元格的列数, 若 Column()则公式在第几列返回第几列的列号

### 2、函数应用实例

#### ①通过身份证号码判断性别

=IF(MOD(RIGHT(LEFT(B2,17),1),2)=0,"女","男")

=IF(MOD(RIGHT(LEFT(B2,17),1),2)=1,"男","女")

或者直接写=IF(MOD(RIGHT(LEFT(B2,17),1),2),"男","女")

原因: 返回值只能是 1 或 0, TRUE 代表 1, FALSE 代表 0

#### ②特殊的舍入方式——员工假期计算

员工假期小数点 $\geq 0.5$ , 假期整天加半天 (0.5 天), 员工假期 $< 0.5$ , 假期就只算整天:

$=IF(MOD(C2,1)\geq 0.5,INT(C2)+0.5,INT(C2))$

$=INT(C2*2)/2$

③基于位置规律的引用

转置使用函数:  $=INDEX(\$A:\$A,COLUMN()-2)$

C7		$=INDEX(\$A:\$A,COLUMN()-2)$											
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	姓名												
2	汪梅												
3	郭磊												
4	林涛		姓名	汪梅	郭磊	林涛	朱健	李明	王建国	陈玉	张华	李丽	汪成
5	朱健												
6	李明												
7	王建国		姓名	汪梅	郭磊	林涛	朱健	李明	王建国	陈玉	张华	李丽	汪成
8	陈玉												
9	张华												
10	李丽												
11	汪成												

跳跃使用函数:

J4		=INDEX(\$E:\$E, ROW()*5-17)									
	A	B	C	D	E	F	G	H	I	J	
1	第一周										
2	H00012769	C014673-004	16	19,269.69	18,982.85		取每周第二个数值 金额				
3	H00012769	C014673-005	40	39,465.17	40,893.08						
4	H00012769	C014673-006	20	21,015.94	22,294.09			40,893.08	3	40893.08	
5	H00012769	C014673-007	20	23,710.26	24,318.37			43,537.56	⊕	43537.56	
6	第二周							51,700.03	13	51700.03	
7	H00012769	C014673-008	16	20,015.07	20,256.69			32,726.66	18	32726.66	
8	H00012769	C014673-009	200	40,014.12	43,537.56			43,537.56	23	43537.56	
9	H00012769	C014673-010	100	21,423.95	22,917.34						
10	H00012769	C014673-011	200	40,014.12	44,258.36						
11	第三周										
12	H00012769	C014673-012	400	84,271.49	92,391.15						
13	H00012769	C014673-013	212	48,705.66	51,700.03						
14	H00012769	C014673-014	224	47,192.03	50,558.50						
15	H00012769	C014673-015	92	21,136.42	22,115.23						
16	第四周										
17	H00012769	C014673-016	100	27,499.51	30,712.18						
18	H00012769	C014673-019	140	29,993.53	32,726.66						
19	H00012774	C015084-001	108	34,682.76	35,738.66						
20	H00012774	C015084-002	72	12,492.95	11,098.92						
21	第五周										
22	H00012769	C014673-008	16	20,015.07	20,256.69						
23	H00012769	C014673-009	200	40,014.12	43,537.56						
24	H00012769	C014673-010	100	21,423.95	22,917.34						
25	H00012769	C014673-011	200	40,014.12	44,258.36						

★首先都需要找规律!!!

第一步写需求, 第二步找规律, 第三步调试, 最后一步写最外层的函数  
分列使用函数:

H1      fx      =INDEX(\$A\$1:\$A\$15, ROW()*3+COLUMN()-10)										
	A	B	C	D	E	F	G	H	I	J
1	汪梅			汪梅	郭磊	林涛		汪梅	2	3
2	郭磊			朱健	李明	王建国		4	5	6
3	林涛			陈玉	张华	李丽		7	8	9
4	朱健			汪成	李军	王红蕾		10	11	12
5	李明			王华	孙传富	赵炎				
6	王建国									
7	陈玉									
8	张华									
9	李丽									
10	汪成									
11	李军									
12	王红蕾									
13	王华									
14	孙传富									
15	赵炎									

## 第十八讲 LOOKUP~浅谈数组

### 1、回顾统计函数

SUMIF & SUMIFS

### 2、认识数组

#### ①数组生成原理

数组也被称为矩阵

例如：=SUM((\$A\$2:\$A\$22=K15)\*(\$B\$2:\$B\$22=L15)\*\$E\$2:\$E\$22)

条件项转换为 0 或 1，然后再去求和

★所有数组长度相等才能运算

★数组函数必须要按住 CTRL+SHIFT 再敲击回车键才能运算，外面会自动加上 { }（这个大括号不能自己手动添加）

★按下 F4 键可以查阅数组的各项具体值，再按下 ESC 键恢复

★数组运算过多会非常慢

#### ②SUMPRODUCT 函数

数组函数，该函数相当于 SUM 函数在数组中运算，{SUM()}

=SUMPRODUCT(((\$A\$2:\$A\$22=K15)\*(\$B\$2:\$B\$22=L15)\*\$E\$2:\$E\$22)

★该函数直接敲回车键运算

### 3、LOOKUP 函数基本应用

#### ①认识 LOOKUP 函数

数组函数

=LOOKUP(lookup\_value, lookup\_vector, [result\_vector])

无精确匹配参数

#### ②LOOKUP 函数模糊匹配

如何使用 LOOKUP 函数精确匹配？

=LOOKUP(1,0/(\$A\$2:\$A\$92=\$G\$2),\$B\$2:\$B\$99)

或=LOOKUP(0,0/(\$A\$2:\$A\$92=\$G\$2),\$B\$2:\$B\$99)

查找项转换为 0 或错误值，然后再去查找，由于 LOOKUP 不找错误值，所以只有一个精确匹配项

★LOOKUP(1,...)原因：由于模糊匹配是用二分法，原理是默认数据源升序排列，和数据源



中间位置的数据比较大，如果查找值较大，则向下继续找中间位置的数据去比较，直到找到目标值，反之则向上查找，对于非数字内容的大小，可以参考升序排序的排列去判断。所以 1 或 0 均可使用

③LOOKUP 函数多条件精确匹配

=LOOKUP(1,(0/(\$A\$2:\$A\$13=I6)\*(\$B\$2:\$B\$13=J6)),\$D\$2:\$D\$13)

或=LOOKUP(0,(0/(\$A\$2:\$A\$13=I6)\*(\$B\$2:\$B\$13=J6)),\$D\$2:\$D\$13)

④使用数组计算个税

=MAX((F5-3500)\*\$C\$4:\$C\$10-\$D\$4:\$D\$10)

G5		{=MAX((F5-3500)*\$C\$4:\$C\$10-\$D\$4:\$D\$10)}					
	A	B	C	D	E	F	G
1							
2		起征点：3500元					
3		全月应纳税所得额(元)	税率	速算扣除数(元)			
4		不超过1500	3%	0		税前月薪	个税
5		1500~4500	10%	105		6500	193
6		4500~9000	20%	555		20000	3120
7		9000~35000	25%	1005		8000	345
8		35000~55000	30%	2755		4000	15
9		55000~80000	35%	5505		12500	1245
10		超过80000	45%	13505			

所求值里面最大的正值就是个税

## 第十九讲 Indirect 函数

### 1、认识 INDIRECT 单元格引用

①了解 indirect 函数的意义及语法  
间接引用

INDIRECT(ref\_text, [a1])

②indirect 函数与 index 函数引用方式的对比

Index：直接引用

=INDEX(E:E,ROW()\*5-25)

=indirect("e"&ROW()\*5-25)

★indirect 函数中固定的值使用“包裹起来”

③处理跨表

提取数据所在位置相同：=INDIRECT(A4&"!G2")

提取数据所在位置的顺序不同：=VLOOKUP("张三",INDIRECT(A4&"!A:G"),7,0)

整张表全部拖拽查找（混合引用）：=VLOOKUP(B\$2,INDIRECT(\$A3&"!\$A:\$G"),7,0)

④跨表引用时的单引号问题

如果在跨表引用时出现错误，可以尝试在代表表名的单元格两边使用英文符号“包裹：

=INDIRECT("'"&A4&"'!G2")

### 2、INDIRECT 名称引用

①为区域定义名称

选中需要定义名称的区域——公式菜单——定义名称——确定

定义名称后，即使写文本，但也能利用 indirect 函数作为引用去对待或运算

★若名称定义错误，可以使用公式菜单——名称管理器——选中错误名称——编辑或删除



## ②制作二级下拉列表

在定义名称之后，可以制作二级下拉列表

数据菜单——数据验证——数据验证——允许：序列——来源：要求具体填写内容——选中需要制作二级下拉列表的单元格——数据验证——数据验证——允许：序列——来源：  
=INDIRECT(F1)

★公式中的 F1 代表的是定义名称之后的单元格区域，例如：在命名区域时，A4:A13 命名为吉林省，B4:B15 命名为江苏省……

A3, B3, C3 都属于一级下拉列表，直接使用数据验证来规定填写内容

	A	B	C
1	产品销售区域		
2			
3	吉林省	江苏省	广东省
4	长春市	南京市	广州市
5	九台市	无锡市	深圳市
6	榆树市	徐州市	珠海市
7	德惠市	常州市	汕头市
8	吉林市	苏州市	佛山市
9	蛟河市	南通市	韶关市
10	桦甸市	连云港市	湛江市
11	舒兰市	淮安市	肇庆市
12	磐石市	盐城市	
13	四平市	扬州市	
14		镇江市	
15		泰州市	

# 第二十讲 图表基础

## 1、认识图表中的元素



图表标题更改：点击图表标题——设置图表标题格式；

选中图表标题——在标题栏里写=号，连接单元格，图表标题会随着单元格的内容而变动

设置坐标轴标题/图例：同上

数据标签：选中图表中的数据，鼠标右键——添加数据标签

模拟运算表：选中图表——图表设计菜单——添加图表元素——数据表——显示图例项标示

坐标轴包括坐标轴、坐标轴标签和主要刻度线，均可进行单独设置

主要网格线：其网格线间隔根据纵坐标轴的单位间隔而决定

## 2、创建并美化柱形图


①了解主次坐标轴的概念

②制作折线于柱形符合图表


有主次纵坐标的图表在 Excel2021 版本中可以直接在更改图表类型中选择组合图!!!

通过调整坐标轴刻度来调整图表在绘图区所在的位置

③制作计划与实际对比图




选中计划完成量的柱子——系列选项中勾选次坐标轴——修改次坐标轴与主坐标轴的刻度一致



④制作双向条形图

先制作簇状条形图——把其中一组数据设置为次要坐标轴——选择次要坐标轴，鼠标右键——设置坐标轴格式——勾选逆序刻度值，最大值为 1，最小值为-1——同样选择主坐标轴，鼠标右键——设置坐标轴格式——最大值为 1，最小值为-1——删除次坐标轴，再次选中主坐标轴——设置坐标轴格式——坐标轴选项——数字——格式代码：0%;0%——添加——选中图片 CTRL+C——点中图表最外层大边框——鼠标右键，设置图表区域格式——填充与线条——图片或纹理填充——剪贴板




### 3、图表通用设置

更改柱形图中柱形的形状：

插入菜单——形状：选择你想要的形状在表格中画出——设置形状格式——CTRL+C——在柱形上直接 CTRL+V

鼠标右键——设置数据系列格式——填充与线条：层叠——可以使用无线条与无填充的矩形框给心形增加空白边框：两者要使用鼠标右键点击组合



保存图表样式：选中图表——鼠标右键——另存为模板

★模板可以通过文件路径导出来，然后在其他电脑中使用

★excel 中的图片更改格式设置：点击图片——鼠标右键：调整图形格式——大小与属性——一点选：随单元格改变位置和大小

## 第二十一讲 经典动态图表实现原理

### 1、动态图表实现原理

①理解图表中的数据系列

数据必须是一列或者一行才能被称为一个系列


②手工修改系列中的数值与坐标轴数据

③小试牛刀——利用 IF 创建简单的动态图表

文件——选项——自定义功能区——勾选开发工具——确定

开发工具菜单——插入——表单控件：直接在表格中使用，简单；ActiveX 控件：在 VBA 中使用的控件，复杂——选中复选框☑——画出复选框——鼠标右键：编辑文字——鼠标右键：设置控件格式——单元格链接——任意一个空单元格：\$G\$2——确定——公式菜单——定义名称——名称：计划完成量；引用位置：=IF(\$G\$2,\$D\$2:\$D\$6,\$F\$2:\$F\$6)（其中 \$D\$2:\$D\$6 为计划完成量数据；\$F\$2:\$F\$6 为空白单元格）——选择空白单元格——插入折线图——选择空白图表——鼠标右键——选择数据——添加——系列名称：计划完成量；系列值：=Sheet1!计划完成量——确定——确定

★其余的复选框图表均可按照这个方法创建



## 2、利用 OFFSET 函数与控件创建动态图表

### ① Offset 函数概述

=OFFSET(reference, rows, cols, [height], [width])

函数是指从一个定位单元格开始，下/上移 n 行，左/右移 n 列，取 n 行 n 列

### ② Offset 函数的动态引用示例

创建不包括表头的数据区域：=OFFSET(\$A\$1,0,0,COUNTA(\$A:\$A),COUNTA(\$1:\$1))

动态的区域，表格里有多少数据就能取多少

定义名称——填写名称和引用位置 (=OFFSET(\$A\$1,0,0,COUNTA(\$A:\$A),COUNTA(\$1:\$1)))


——插入菜单——数据透视表——数据透视图和数据透视表——表/区域：填写刚刚命名的名称——选择放置数据透视表的位置：新工作表——确定

### ③ 动态图表 1 永远返回最后 10 行数据

表明我们需要取该列最后一个数据的前十行的单元格，例如：数据 19 行，从第 9 行下面取十行，数据 21 行从第 11 列下面取十行，均为最后一个单元格所在位置-10

定义名称——引用位置：=OFFSET(\$B\$1,COUNTA(\$B:\$B)-10,0,10,1)——画出图表——选择数据——添加——系列值：=sheet1!营业额——确定——确定

★ 需要按照同样的方法，确定横坐标显示的标签：定义名称——引用位置：=OFFSET(\$A\$1,COUNTA(\$A:\$A)-10,0,10,1)——画出图表——选择数据——编辑——轴标签区域：=sheet1!类型——确定——确定



### ④ 动态图表 2 通过控件控制图表数据


第一个滚动条：控制数据向下滚动

第二个滚动条：控制数据在图表中显示的条数

开发工具菜单——插入——表单控件：滚动条（窗体控件）——选中滚动条——鼠标右键

——设置控件格式——控制——最小值：1；单元格链接：\$D\$2——确定——另外一条滚动条链接到\$D\$4——公式菜单——定义名称——名称：营业；引用位置：  
=OFFSET(\$B\$1,\$D\$2,0,\$D\$4,1)——确定——插入菜单——折线图——选择数据——添加——确定——确定

★需要按照同样的方法，确定横坐标显示的标签：定义名称——引用位置：  
=OFFSET(\$A\$1,\$D\$2,0,\$D\$4,1)——画出图表——选择数据——编辑——轴标签区域：  
=sheet1!组别——确定——确定



## 第二十二讲 制作甘特图与动态甘特图

### 1、制作甘特图

甘特图：用来进行时间或任务管理

选中数据——堆积条形图——隐藏日期开始时间的条形区域，仅保留时间间隔的条形——  
选中坐标轴，鼠标右键设置坐标轴格式——修改坐标轴最大值和最小值，让条形图顶着纵坐标



### 2、制作动态甘特图

将天数拆分为两列，分别为已完成天数和未完成天数

已完成天数有三种情况：当天<计划开始日，完成天数为 0，未完成天数为原计划天数；当天>计划结束日（计划开始日+计划天数），完成天数为计划天数，未完成天数为 0；当天在计划开始与结束日之间，完成天数为当天—计划开始日。

未完成天数：计划天数—已完成天数

已完成公式：=IF(\$B\$11<B2,0,IF(\$B\$11>B2+C2,C2,\$B\$11-B2))

未完成公式：=C2-E2

选中 A、B、E、F 四列做堆积条形图——修改图表样式——开发工具菜单——插入——滚

动条——选中滚动条，鼠标右键——设置控件格式——最小值为 0，最大值为计划天数的总和——单元格链接到空白单元格\$C\$11——确定——B11 单元格（今天的具体日期）写：  
 $=44745$ （计划开始日）+C11——插入菜单——文本——文本框——绘制文本框，点中——在编辑栏上写： $=B$11$




## 第二十三讲 饼图美化与 PPT 图表

### 1、双坐标柱形图的补充知识

#### ①主次坐标设置


#### ②主次坐标柱形避让

若制作柱形图设置主次坐标之后出现了柱形的重叠，而我们想让它们呈现平铺摆放的样式，则可以：选中图表——鼠标右键，选择数据——添加——共添加两个系列，系列值均为： $=\{0\}$ ——确定——格式菜单——当前所选内容：系列三——设置所选内容格式——系列选项：勾选主坐标轴——再次选中条形，鼠标右键选择数据——在图例项中将系列三提到最前面——确定



### 2、饼图美化

选中饼图——鼠标右键——三维旋转——取消自动缩放——调整高度（原始高度百分比）——鼠标右键，设置数据系列格式——效果——三维格式——棱台效果设置——阴影




双层饼图：选中需要放在前面的饼图的数据——插入二维饼图——选中饼图——鼠标右



键，选择数据——添加——系列值：仅仅只选择数据，不选择图例（分类）的单元格——确定——格式菜单——当前所选内容——选择需要放置在后面的饼图——设置所选内容格式——系列选项：勾选次坐标轴——选中全部的前置的饼图向外拉动——再单独的一块块的拖拽到圆心位置拼接起来

★哪个饼图放在前面就要先制作哪个饼图



### 3、PPT 中的图表

#### ①Excel 图表在 PPT 中粘贴

PPT 更改默认图表/字体/图形的颜色：设计选项卡——颜色（在变体的下拉框里面）——更改主题色

如果出现图表粘贴在 PPT 中颜色设置不一致的情况，表明 PPT 的默认颜色与 Excel 不同，需要在粘贴选项中选择保留源格式和嵌入工作簿

#### ②Excel 图表在 PPT 中更新

在粘贴选项中选择保留源格式和链接数据

在 Excel 中更改数据之后，需要在图表设计选项卡中刷新数据

在每次打开 PPT 时会自动询问是否要更新链接：开始选项卡的粘贴——选择性粘贴——粘贴链接——确定

#### ③PPT 制作动态柱形图（柱子一根根连续出现）

选中图表——点击 PPT 动画选项卡——动画窗格——添加动画——点中动画窗格中的动作——鼠标右键，效果选项——图表动画——组合图表：按系列——确定——选中系列的全部动作——鼠标右键：从上一项之后开始

## 第二十四讲 宏表函数

### 1、利用宏表函数获取信息

#### ①get.cell 函数

GET.CELL(Type\_num, Reference)

Type\_num 指明单元格中信息的类型。用数字表示，范围为 1-66。

Reference 为引用的单元格或区域。

★宏表函数的使用方法：选中空白单元格——公式选项卡——定义名称——名称：计算颜色——引用位置：=get.cell(63,A2)——确定——在空白单元格里写：=计算颜色

名称一定要命名！而且必须要在与 A2 对应的 B2 单元格中填写该公式，位置改变会影响计



算!

63: 代表返回单元格填充的背景色所代表的数字

24: 代表返回单元格字体颜色所代表的数字

6: 表示获取该单元格的公式; 也可以使用 FORMULATEXT()函数

②get.workbook 函数

获取整个文件信息

GET.WORKBOOK(信息类型, 工作簿名字)

1: 返回工作簿中所有工作表的名字

3: 返回工作簿中当前选择表的名称

4: 返回工作簿中工作表的个数

38: 返回活动工作表的名字

同样在定义名称中使用该函数, 如=GET.WORKBOOK(1)——选中空白单元格——公式:

=INDEX(工作表名,row())

给每个工作表生成超链接: =HYPERLINK(INDEX(工作表名, ROW())&"!A1")

★=HYPERLINK([表名]!A1): 超链接必须链接到某一个单元格

## 2、宏表函数常见应用

①EVALUATE 函数

EVALUATE(formula\_text), 该函数可以将文本转化为数组

公式选项卡——定义名称——名称: 运算; 引用位置: =EVALUATE(A3)——在空白单元格中写: =运算

★SUBSTITUTE 函数: SUBSTITUTE(text, old\_text, new\_text, [instance\_num])

=SUBSTITUTE(A9, " ", "+"); 替换连字符, 强大之处在于可以指定替换第几个连字符

★若一个单元格出现 89, 90, 78, 需要把数据转化为数组进行求和运算: =SUM(计算);

计算: =EVALUATE("{&A9&}")